Differential Equations

Scrierea ecuatiilor diferentiale ordinare (EDO) in Mathematica

Sa presupunem urmatoarea EDO:

$$\frac{d^2y}{dx^2} + 6\frac{dy}{dx} + 13y^2 = 0$$

In *Mathematica* putem reprezenta aceasta ecuatie astfel:

```
EDO1 = y''[x] + 6 y'[x] + 13 y[x]^2 == 0
```

```
13 y[x]^2 + 6 y'[x] + y''[x] == 0
```

Functia **Equal** (==) este folosita pentru a desemna o relatie logica intre expresiile din stanga si dreapta acestuia.

Un alt mod de a scrie o EDO consta in utilizarea functiei **D**

```
EDO2 = D[y[x], {x, 2}] + 6 D[y[x], x] + 13 y[x]^2 == 0
```

```
13 y[x]^2 + 6 y'[x] + y''[x] == 0
```

Functia de baza DSolve

Functia utilixata in rezolvarea EDO este **DSolve[*eqns*, *y[x]*, *x*].** Primul argument al lui**DSolve** este o ecuatie sau o lista de ecuatii, cel de-al doilea este variabila dependenta sau o lista de variabile, iar cel de-al treilea este variabila independenta. Ne propunem sa rezolvam myEDO1:

```
DSolve[EDO1, y[x], x]
```

```
DSolve[13 y[x]^2 + 6 y′[x] + y″[x] == 0, y[x], x]
```

Nu se intampla nimic deoareca ecuatia EDO1 este neliniara.

Sa incercam o alta ecuatie:

```
EDO2 = y''[x] + 6 y'[x] + 13 y[x] == 0;
Sol2 = DSolve[EDO2, y[x], x]
```

```
{{y[x] → e^(-3 x) C[2] Cos[2 x] + e^(-3 x) C[1] Sin[2 x]}}
```

In acest caz *Mathematica* este capabila sa gaseasca solutia simbolica a ec.EDO2. Sa nu uitam ca solutia este data in termenii regulii de inlocuire pentru **y[x]**. **C[1]**, **C[2]** sunt constante de integrare. Pentru a obtine forma explicita a luir y[x] utilizam functia **ReplaceAll.**

```
y[x] /. Sol2
```

```
{e^(-3 x) C[2] Cos[2 x] + e^(-3 x) C[1] Sin[2 x]}
```

Cum pot fi incluse conditiile initiale in solutia ecuatiei diferentiale ?

Fie ecuatia diferentiala:

-y′[x] + y″[x] == 0

cu conditiile initiale

y[0]==1,y'[0]==2

```
ecder = D[y[x], {x, 2}] - D[y[x], x] == 0
```

```
-y′[x] + y″[x] == 0
```

```
soln = DSolve[ecder, y, x]
```

```
{{y → Function[{x}, e^x C[1] + C[2]]}}
```

```
soln[[1]][[1]]
```

```
y → Function[{x}, ⅇ^x C[1] + C[2]]
```

```
soln[[1]][[1]] /. x -> 0
```

Function::flpar :
Parameter specification {0} in Function[{0}, e^0 C[1] + C[2]] should be a symbol or a list of symbols. ≫

```
y → Function[{0}, ⅇ^0 C[1] + C[2]]
```

```
D[y[x], x] == 2 /. soln[[1]][[1]] /. x -> 0
```

```
C[1] == 2
```

O metoda este

```
eq1 = {y[x] == 1 /. soln[[1]][[1]] /. x -> 0,
  D[y[x], x] == 2 /. soln[[1]][[1]] /. x -> 0}
```

```
{C[1] + C[2] == 1, C[1] == 2}
```

```
Solve[eq1]
```

```
{{C[1] → 2, C[2] → -1}}
```

O alta metoda este cea data de Mathematica :

```
DSolve[{ecder, y[0] == 1, y'[0] == 2}, y[x], x]
```

```
{{y[x] → -1 + 2 ⅇ^x}}
```

Daca avem un set de conditii initiale, spre exemplu:

y[0] = 0, y'[0] = 1

le putem folusi direct in **DSolve** pentru a obtine solutia unica. Conditiile initiale sunt adunate la primul argument al lui **DSolve** pentru a forma o lista de ecuatii:

```
Solunica = DSolve[{y''[x] + 6 y'[x] + 13 y[x] == 0, y[0] == 0, y'[0] == 1},
  y[x], x]
```

$$\left\{\left\{y[x] \to \frac{1}{2} e^{-3x} \mathrm{Sin}[2x]\right\}\right\}$$

Din nou solutia este data ca o regula de inlocuire pentru **y[x]**. Daca dorim tiparirea solutiei utilizam functia **Plot**.Nu putem utiliza iesirea din **DSolve** in **Plot** (deoarece iesirea nu este o functie ci mai degraba o regula de inlocuiree). Iesirea din impas se poate realiza cu ajutorul functiei **ReplaceAll** care da forma explicita a solutiei:

```
Plot[y[x] /. Solunica, {x, 0, 3}, PlotStyle → RGBColor[0, 0, 1],
 AxesLabel → {"x", "y(x)"}]
```



Rutina **DSolve** poate rezolva si probleme cu valori limita liniare. Spre exemplu, daca conditiile la limita sunt

y[0] = 0, y[1] = 1

Avem

```
Solunica = DSolve[{(y′)′[x] + 6 y′[x] + 13 y[x] == 0, y[0] == 0, y[1] == 1},
  y[x], x]
Plot[y[x] /. Solunica, {x, 0, 1}, PlotStyle → RGBColor[0, 0, 1],
 AxesLabel → {"x", "y(x)"}]
```

$\{\{y[x] \to e^{3-3x} \, Csc[2] \, Sin[2x]\}\}$



Familia solutiilor unei EDO

In anumite aplicatii se cere studiul familiei solutiilor unei EDO. Prin aceasta se intelege determinarea modului de variatie al solutiilor in raport cu constantele de integrare. In primul rand trebuie vazut daca EDo are sau nu o constanta de integrare

```
Solunica4 = DSolve[y'[x] == Exp[-x] - y[x], y[x], x]
```

$\{\{y[x] \to e^{-x} x + e^{-x} C[1]\}\}$

Studiem modul de variatie al solutiei in functie de valorile lui C[1]. Deci, vom creea cate o solutie particulara pentru fiecare valoare a lui C[1], dupa care le vom reprezenta grafic intr-o singura diagrama.Pentru a genera lista solutiilor vom folosi functia **Table** impreuna cu **ReplaceAll.**

```
Table[y[x] /. Solunica4 /. C[1] -> i, {i, 1, 4}]
```

```
{{e^-x + e^-x x}, {2 e^-x + e^-x x}, {3 e^-x + e^-x x}, {4 e^-x + e^-x x}}
```

In primul rand vom inlocui y[x] cu solutia dorita folosindu-ne de regula generata de **DSolve**. In al doilea rand vom folosi pasii din Table pentru valorile lui i pentru a utiliza regula de inlocuire a lui C[1] cu valoarea dorita. Incapsuland regula pentru C[1] in **Table** se va genera in mod automat familia de solutii.

```
Plot[Evaluate[Table[y[x] /. Solunica4 /. C[1] → i, {i, 1, 4}]],
 {x, 0, 3},
 PlotStyle → {RGBColor[0, 0, 1], RGBColor[0.5`, 0.5`, 0],
   RGBColor[0.4`, 0, 0.6`], RGBColor[0.6`, 0, 0.4`]}]
```



Probleme cu valori proprii

Presupunem ca dorim sa rezolvam o problema de valori proprii cu *Mathematica* .

$$\phi'' + \lambda\phi = 0$$

BC1: $\phi'[1] = 0$

BC2: $\phi[0] = 0$

Datorita faptului ca exista intotdeauna o solutie trivialan\ nu putem folosi DSolve

```
DSolve[{ϕ''[s] + λ ϕ[s] == 0, ϕ'[1] == 0, ϕ[0] == 0}, ϕ[s], s]
```

```
{{ϕ[s] → 0}}
```

in primul rand, vom gasi solutia generala a EDO si o vom exprima ca o functie pura. Aceasta cere specificarea celui de-al doilea argument din DSolve ca ϕ si nu ca ϕ[s].

```
genSol = First[DSolve[ϕ''[s] + λ ϕ[s] == 0, ϕ, s]]
```

$$\left\{\phi \to \text{Function}\left[\{s\},\ C[1]\ \text{Cos}\left[s\sqrt{\lambda}\right] + C[2]\ \text{Sin}\left[s\sqrt{\lambda}\right]\right]\right\}$$

Apoi, vom determina conditiile la limita in termenii solutiei generale

```
BC1 = (ϕ'[1] == 0) /. genSol
```

$$\sqrt{\lambda}\ C[2]\ \text{Cos}\left[\sqrt{\lambda}\right] - \sqrt{\lambda}\ C[1]\ \text{Sin}\left[\sqrt{\lambda}\right] == 0$$

```
BC2 = (ϕ[0] == 0) /. genSol
```

```
C[1] == 0
```

Solve nu se poate utiliza in determinarea valorilor proprii deoarece solutia va fi triviala (cum se poate vedea mai jos)

```
Solve[{BC1, BC2}, {C[1], C[2]}]
```

```
{{C[2] → 0, C[1] → 0}}
```

Trucul consta in determinarea determinantului matricii coeficien-tilor cu ajutorul functiei **Coefficient**, cuplata cu **Map**[*f*, *expr*] sau ***f* /@ *expr*** care aplica *f* fiecarui element din *expr*.

```
A = Map[Coefficient[First[#], {C[1], C[2]}] &, {BC1, BC2}]
```

$$\left\{\left\{-\sqrt{\lambda}\ \mathrm{Sin}\left[\sqrt{\lambda}\right], \sqrt{\lambda}\ \mathrm{Cos}\left[\sqrt{\lambda}\right]\right\}, \{1, 0\}\right\}$$

Avem deci o matrice 2×2 .Ecuatia caracteristica pentru valorile proprii se determina prin egalarea cu zero a determinantului sau.

```
charEqn = (Det[A] // Simplify) == 0
```

$$-\sqrt{\lambda}\ \mathrm{Cos}\left[\sqrt{\lambda}\right] == 0$$

Aceasta fiind o ecuatie transcendentala in λ functia **Solve** nu este capabila sa determine toate radacinile acestei ecuatii (dupa cum se poate vedea)

```
Solve[charEqn, λ]
```

Solve::ifun :
Inverse functions are being used by Solve, so some solutions may not be found; use Reduce for complete solution information. ≫

$$\left\{\{\lambda \to 0\}, \left\{\lambda \to \frac{\pi^2}{4}\right\}\right\}$$

Exista posibilitatea crearii unei functii in *Mathematica* care genere-aza radacini

$$\lambda_{n\_} := (2\,n + 1)^2 \frac{\pi^2}{4} \ /;\ n > 0$$

$$\lambda_{n\_} := 0 \ /;\ n == 0$$

Lista radacinilor va fi

```
Table[λn, {n, 0, 10}]
```

$$\left\{0, \frac{9\pi^2}{4}, \frac{25\pi^2}{4}, \frac{49\pi^2}{4}, \frac{81\pi^2}{4}, \frac{121\pi^2}{4}, \frac{169\pi^2}{4}, \frac{225\pi^2}{4}, \frac{289\pi^2}{4}, \frac{361\pi^2}{4}, \frac{441\pi^2}{4}\right\}$$

Urmatorul pas consta in determinarea coeficientilor C[1] si C[2].
In acest caz : C[2]=0, iar C[1] este arbitrar. Astfel functiile proprii sunt

$\phi_n[s] = \text{Cos}\left[\sqrt{\lambda_n}\ s\right]$, n=0,1,2,...

Deci in *Mathematica*

```
ϕn_[s_] := Sin[Sqrt[λn] s]
```

```
ϕ2[s]
```

$$\text{Sin}\left[\frac{5\pi s}{2}\right]$$

Putem usor demonstra ortogonalitatea acestor functii proprii

```
Map[∫_0^1 ϕ3[s] ϕ#[s] ds &, Range[0, 10]]
```

$$\left\{0, 0, 0, \frac{1}{2}, 0, 0, 0, 0, 0, 0, 0\right\}$$

Aplicatia 1

Utilizand DSolve determinati solutiile urmatoarelor ecuatii diferentiale:

(i) $\frac{dy}{dt} + 4\,y = t^2$

(ii) $\frac{d^2y}{dt^2} + 4\,\frac{dy}{dt} = t^2$

(iii) $\frac{d^2y}{dt^2} - 4\,\frac{dy}{dt} + 3\,y = 0$

(iv) $\frac{d^2y}{dt^2} - 4\,\frac{dy}{dt} + 3\,y = t^2$

```
DSolve[y'[t] + 4 y[t] == t^2, y[t], t]
```

$$\left\{\left\{y[t] \to \frac{1}{32}\left(1 - 4\,t + 8\,t^2\right) + e^{-4\,t}\,C[1]\right\}\right\}$$

```
DSolve[y''[t] + 4 y'[t] == t^2, y[t], t]
```

$$\left\{\left\{y[t] \to \frac{t}{32} - \frac{t^2}{16} + \frac{t^3}{12} - \frac{1}{4}\,e^{-4\,t}\,C[1] + C[2]\right\}\right\}$$

```
DSolve[y''[t] - 4 y[t] + 3 y[t] == 0, y[t], t]
```

$$\left\{\left\{y[t] \to e^{t}\,C[1] + e^{-t}\,C[2]\right\}\right\}$$

```
DSolve[y''[t] - 4 y[t] + 3 y[t] == t^2, y[t], t]
```

$$\{\{y[t] \to -2 - t^2 + e^t C[1] + e^{-t} C[2]\}\}$$

Aplicatia 2

Fie urmatoarele ecuatii diferentiale

(i) $\frac{d^2y}{dx^2} + \frac{dy}{dx} + \frac{5}{4} y = 0$

(ii) $\frac{d^2y}{dx^2} + 2 \frac{dy}{dx} + \frac{1}{4} y = 0$

Utilizati DSolve pentru a determina solutia generala.

```
DSolve[y''[t] + y'[t] + 5/4 y[t] == 0, y[t], t]
```

$$\{\{y[t] \to e^{-t/2} C[2] \operatorname{Cos}[t] + e^{-t/2} C[1] \operatorname{Sin}[t]\}\}$$

```
DSolve[y''[t] + 2 y[t] + 1/4 y[t] == 0, y[t], t]
```

$$\left\{\left\{y[t] \to C[1] \operatorname{Cos}\left[\frac{3 t}{2}\right] + C[2] \operatorname{Sin}\left[\frac{3 t}{2}\right]\right\}\right\}$$

Aplicatia 3

Foloditi DSolve pentru a rezolva

(i) $\frac{d^2y}{dt^2} + 5y = t\,\mathrm{Sin}(t),\ y(0) = 1,\ \frac{dy}{dt}(0) = \pi$

Reprezentati grafic solutia pentru 0<t<6π.

(ii) $\frac{d^2y}{dt^2} + 5y = t\,\mathrm{Sin}(t),\ y(0) = 1,\ \frac{dy}{dt}(2\pi) = 0$

Reprezentati grafic solutia pentru 0<t<2π

```
sol = DSolve[{y''[t] + 5 y[t] == t Sin[t], y[0] == 1, y'[0] == π},
    y[t], t]
```

$$\left\{\left\{y[t] \to \frac{1}{40}\left(45\,\mathrm{Cos}\left[\sqrt{5}\;t\right] - 5\,\mathrm{Cos}[t]\,\mathrm{Cos}\left[\sqrt{5}\;t\right]^2 + 10\,t\,\mathrm{Cos}\left[\sqrt{5}\;t\right]^2\mathrm{Sin}[t] + 8\sqrt{5}\,\pi\,\mathrm{Sin}\left[\sqrt{5}\;t\right] - 5\,\mathrm{Cos}[t]\,\mathrm{Sin}\left[\sqrt{5}\;t\right]^2 + 10\,t\,\mathrm{Sin}[t]\,\mathrm{Sin}\left[\sqrt{5}\;t\right]^2\right)\right\}\right\}$$

```
sol // Simplify
```

$$\left\{\left\{y[t] \to -\frac{\mathrm{Cos}[t]}{8} + \frac{9}{8}\mathrm{Cos}\left[\sqrt{5}\;t\right] + \frac{1}{4}\,t\,\mathrm{Sin}[t] + \frac{\pi\,\mathrm{Sin}\left[\sqrt{5}\;t\right]}{\sqrt{5}}\right\}\right\}$$

```
Plot[y[t] /. sol, {t, 0, 2 π},
  PlotStyle → RGBColor[0.6`, 0, 0.4`]]
```




```
sol1 =
  DSolve[{y''[t] + 5 y[t] == t Sin[t], y[0] == 1, y'[2 π] == 0},
   y[t], t]
```

$$\left\{\left\{y[t] \to \frac{1}{40}\left(45\,\mathrm{Cos}\left[\sqrt{5}\;t\right] - 5\,\mathrm{Cos}[t]\,\mathrm{Cos}\left[\sqrt{5}\;t\right]^2 + 10\,t\,\mathrm{Cos}\left[\sqrt{5}\;t\right]^2 \mathrm{Sin}[t] - 4\sqrt{5}\,\pi\,\mathrm{Cos}\left[2\sqrt{5}\,\pi\right]\mathrm{Sin}\left[\sqrt{5}\;t\right] - 5\,\mathrm{Cos}[t]\,\mathrm{Sin}\left[\sqrt{5}\;t\right]^2 + 10\,t\,\mathrm{Sin}[t]\,\mathrm{Sin}\left[\sqrt{5}\;t\right]^2 + 45\,\mathrm{Sin}\left[\sqrt{5}\;t\right]\mathrm{Tan}\left[2\sqrt{5}\,\pi\right] - 4\sqrt{5}\,\pi\,\mathrm{Sin}\left[2\sqrt{5}\,\pi\right]\mathrm{Sin}\left[\sqrt{5}\;t\right]\mathrm{Tan}\left[2\sqrt{5}\,\pi\right]\right)\right\}\right\}$$

```
sol1 // Simplify
```

$$\left\{\left\{y[t] \to -\frac{1}{80}\,\mathrm{Sec}\left[2\sqrt{5}\,\pi\right]\left(5\,\mathrm{Cos}\left[2\sqrt{5}\,\pi - t\right] - 90\,\mathrm{Cos}\left[\sqrt{5}\,(-2\pi + t)\right] + 5\,\mathrm{Cos}\left[2\sqrt{5}\,\pi + t\right] + 10\,t\,\mathrm{Sin}\left[2\sqrt{5}\,\pi - t\right] + 8\sqrt{5}\,\pi\,\mathrm{Sin}\left[\sqrt{5}\,t\right] - 10\,t\,\mathrm{Sin}\left[2\sqrt{5}\,\pi + t\right]\right)\right\}\right\}$$

```
Plot[y[t] /. sol1, {t, 0, 2 π},
  PlotStyle → RGBColor[0, 0, 1],
  AxesLabel → {"t", "y(t)"}]
```



Aplicatia 4

Fie EDO:

$$x \frac{dy}{dx} - \left(2 + x \operatorname{Log}\left[\frac{x^2}{y[x]}\right]\right) y[x] = 0$$

Utilizati DSolve pentru a gasi solutia.

Verificati daca solutia gasita este corecta.

```
Clear["Global`*"]
```

```
deq1 = x D[y[x], x] == (x Log[x ^ 2 / y[x]] + 2) y[x]
```

$$x\, y'[x] == \left(2 + x \operatorname{Log}\left[\frac{x^2}{y[x]}\right]\right) y[x]$$

```
sol1 = DSolve[deq1, y[x], x]
```

$$\left\{\left\{y[x] \to e^{-2 e^{-x} C[1]} x^2\right\}\right\}$$

```
s1 = x D[y[x] /. sol1[[1]][[1]], x]
```

$$x \left(2 e^{-2 e^{-x} C[1]} x + 2 e^{-x-2 e^{-x} C[1]} x^2 C[1]\right)$$

```
d1 = (x Log[x ^ 2 / y[x]] + 2) y[x] /. sol1[[1]][[1]]
```

$$e^{-2 e^{-x} C[1]} x^2 \left(2 + x \operatorname{Log}\left[e^{2 e^{-x} C[1]}\right]\right)$$

```
Simplify[s1 - d1]
```

$$e^{-x-2 e^{-x} C[1]} x^3 \left(2 C[1] - e^x \operatorname{Log}\left[e^{2 e^{-x} C[1]}\right]\right)$$

```
s1 - d1 /. Log[Exp[f_]] -> f // Simplify
```

```
0
```

Fie EDO:

$y'' + \text{sign}(x)\, y = 0;\ y(0) = 1,\ y'(0) = 0.$

Utilizati DSolve pentru a gasi solutia.

Verificati daca solutia gasita este corecta.

```
DSolve[{y''[x] - y[x] == 0, y[0] == 1, y'[0] == 0}, y[x], x] //
  FullSimplify
```

```
{{y[x] → Cosh[x]}}
```

```
DSolve[{y''[x] + y[x] == 0, y[0] == 1, y'[0] == 0}, y[x], x] //
  FullSimplify
```

```
{{y[x] → Cos[x]}}
```

OSCILATORUL LINIAR

■ REZOLVARE

```
Clear["Global`*"]
```

■ a) $\frac{d^2 x}{dt^2} + \omega \cdot x = 0$

$x(0) = x_0;\ v(0) = v_0$

```
eq1 = x''[t] + ω0^2 x[t] == 0 ;
```

```
initial1 = {x[0] == x0, x′[0] == v0};
```

Flatten[*list*] netezeste listele grupate"Flatten[{a,{b,c},{d}}] ⟶ {a, b, c, d}"

```
dSol1 = DSolve[{eq1 , initial1} // Flatten, x[t], t] //
   Flatten
```

$$\left\{x[t] \to \frac{x0\,\omega 0\,\mathrm{Cos}[t\,\omega 0] + v0\,\mathrm{Sin}[t\,\omega 0]}{\omega 0}\right\}$$

```
Needs["Graphics`Arrow`"]
```

General::obspkg :
Graphics`Arrow` is now obsolete. The legacy version being loaded may conflict with current *Mathematica* functionality. See the Compatibility Guide for updating information. ≫

```
?Graphics`Arrow`*
```

▼ Graphics`Arrow`

Absolute	HeadCenter	HeadLength	HeadScaling	HeadShape	HeadWidth	Relative	ZeroShape

```
?Arrow
```

Arrow[{pt_1, pt_2}] is a graphics primitive which represents an arrow from pt_1 to pt_2.
Arrow[{pt_1, pt_2}, s] represents an arrow with its ends set back from pt_1 and pt_2 by a distance s.
Arrow[{pt_1, pt_2}, {s_1, s_2}] sets back by s_1 from pt_1 and s_2 from pt_2. ≫

```
arrow1 = {
        {Hue[0.2], Arrow[{0, 0}, {0.5, 0.5}]},
        {Hue[0.4], Arrow[{0, 0}, {2, 0}]},
        {Hue[0.6], Arrow[{0, 0}, {0, 0.5}]}
    };
```

```
Plot[x[t] /. dSol1 /. {x0 → 0, ω0 → 1, v0 → 1},
  {t, -2 π, 2 π}, Epilog → arrow1]
```



- **b) x(0)=A Cos[δ]; v(0)=-A ω0 Sin[δ]**

```
initial2 = {x[0] == A Cos[δ], x'[0] == -A ω0 Sin[δ]};
```

```
dSol2 =
  DSolve[{eq1, initial2} // Flatten, x[t], t] // Flatten //
    Simplify
```

```
{x[t] → A Cos[δ + t ω0]}
```

c)

```
lapTx = LaplaceTransform[eq1, t, s]
```

$$s^2\,\text{LaplaceTransform}[x[t], t, s] + \omega0^2\,\text{LaplaceTransform}[x[t], t, s] - s\,x[0] - x'[0] == 0$$

```
lapSol = Solve[lapTx, LaplaceTransform[x[t], t, s]][[1]]
```

$$\left\{\text{LaplaceTransform}[x[t], t, s] \to \frac{s\,x[0] + x'[0]}{s^2 + \omega0^2}\right\}$$

Apart[*expr*] rescrie o expresie rationala ca o suma de termeni cu numar minim de numitori

Apart[(x^2+1)/(x-1)]⟶1 + 2 / (-1 + x) + x

Apart de descompunerea fractionala partiala a unei expresii rationale

```
InverseLaplaceTransform[
    LaplaceTransform[x[t], t, s] /. lapSol, s, t] //
  Apart
```

$$\text{Cos}[t\,\omega0]\,x[0] + \frac{\text{Sin}[t\,\omega0]\,x'[0]}{\omega0}$$

Solve

y[t] + 5 y′[t] + y′′[t] == Cos[t] − DiracDelta[−2 π + t]

In[1]:=
```
Clear[y]
```

In[2]:=
```
L[y_] := D[y, {t, 2}] + 5 D[y, {t, 1}] + y == Cos[t] - DiracDelta[t - 2 Pi]
```

In[3]:=
```
L[y[t]]
```

Out[3]=

$$y[t] + 5\,y'[t] + y''[t] == \mathrm{Cos}[t] - \mathrm{DiracDelta}[-2\pi + t]$$

In[4]:=
```
soleq = DSolve[{L[y[t]], y[0] == 0, y'[0] == 0}, y[t], t]
```

Out[4]=

$$\Big\{\Big\{y[t] \to -\frac{1}{105\left(-5+\sqrt{21}\right)\left(5+\sqrt{21}\right)} 2\, e^{-\frac{1}{2}\left(-5+\sqrt{21}\right)(2\pi+t)} \Big(2\sqrt{21}\, e^{\left(-\frac{5}{2}-\frac{\sqrt{21}}{2}\right)t+\frac{1}{2}\left(-5+\sqrt{21}\right)(2\pi+t)} - 2\sqrt{21}\, e^{\left(-\frac{5}{2}+\frac{\sqrt{21}}{2}\right)t+\frac{1}{2}\left(-5+\sqrt{21}\right)(2\pi+t)} - 2\sqrt{21}\, e^{\frac{1}{2}\left(-5+\sqrt{21}\right)(2\pi+t)}\mathrm{Cos}[t] + 2\sqrt{21}\, e^{\left(-5+\sqrt{21}\right)\pi+\left(-\frac{5}{2}+\frac{\sqrt{21}}{2}\right)t}\mathrm{Cos}[t] - 10\sqrt{21}\, e^{2\left(-\frac{5}{2}+\frac{\sqrt{21}}{2}\right)t}\mathrm{HeavisideTheta}[-2\pi+t] + 10\sqrt{21}\, e^{\left(5+\sqrt{21}\right)\pi+\left(-\frac{5}{2}-\frac{\sqrt{21}}{2}\right)t+\frac{1}{2}\left(-5+\sqrt{21}\right)(2\pi+t)}\mathrm{HeavisideTheta}[-2\pi+t] + 21\, e^{\frac{1}{2}\left(-5+\sqrt{21}\right)(2\pi+t)}\mathrm{Sin}[t] - 5\sqrt{21}\, e^{\frac{1}{2}\left(-5+\sqrt{21}\right)(2\pi+t)}\mathrm{Sin}[t] + 21\, e^{\left(-5+\sqrt{21}\right)\pi+\left(-\frac{5}{2}+\frac{\sqrt{21}}{2}\right)t}\mathrm{Sin}[t] + 5\sqrt{21}\, e^{\left(-5+\sqrt{21}\right)\pi+\left(-\frac{5}{2}+\frac{\sqrt{21}}{2}\right)t}\mathrm{Sin}[t]\Big)\Big\}\Big\}$$

In[5]:=
```
sol = y[t] /. soleq[[1]]
```

Out[5]=

$$-\frac{1}{105\left(-5+\sqrt{21}\right)\left(5+\sqrt{21}\right)} 2\, e^{-\frac{1}{2}\left(-5+\sqrt{21}\right)(2\pi+t)} \Big(2\sqrt{21}\, e^{\left(-\frac{5}{2}-\frac{\sqrt{21}}{2}\right)t+\frac{1}{2}\left(-5+\sqrt{21}\right)(2\pi+t)} - 2\sqrt{21}\, e^{\left(-\frac{5}{2}+\frac{\sqrt{21}}{2}\right)t+\frac{1}{2}\left(-5+\sqrt{21}\right)(2\pi+t)} - 2\sqrt{21}\, e^{\frac{1}{2}\left(-5+\sqrt{21}\right)(2\pi+t)}\mathrm{Cos}[t] + 2\sqrt{21}\, e^{\left(-5+\sqrt{21}\right)\pi+\left(-\frac{5}{2}+\frac{\sqrt{21}}{2}\right)t}\mathrm{Cos}[t] - 10\sqrt{21}\, e^{2\left(-\frac{5}{2}+\frac{\sqrt{21}}{2}\right)t}\mathrm{HeavisideTheta}[-2\pi+t] + 10\sqrt{21}\, e^{\left(5+\sqrt{21}\right)\pi+\left(-\frac{5}{2}-\frac{\sqrt{21}}{2}\right)t+\frac{1}{2}\left(-5+\sqrt{21}\right)(2\pi+t)}\mathrm{HeavisideTheta}[-2\pi+t] + 21\, e^{\frac{1}{2}\left(-5+\sqrt{21}\right)(2\pi+t)}\mathrm{Sin}[t] - 5\sqrt{21}\, e^{\frac{1}{2}\left(-5+\sqrt{21}\right)(2\pi+t)}\mathrm{Sin}[t] + 21\, e^{\left(-5+\sqrt{21}\right)\pi+\left(-\frac{5}{2}+\frac{\sqrt{21}}{2}\right)t}\mathrm{Sin}[t] + 5\sqrt{21}\, e^{\left(-5+\sqrt{21}\right)\pi+\left(-\frac{5}{2}+\frac{\sqrt{21}}{2}\right)t}\mathrm{Sin}[t]\Big)$$

In[6]:=
```
realsolplot = Plot[sol, {t, 0, 20}, PlotStyle → {Red, Thick}]
```

Out[6]=



We can use this (mainly the plot) as a check if we like.